

Oscilaciones amortiguadas y forzadas

■ Oscilador armónico amortiguado

Solución general

Recuerde que la segunda ley de Newton ($\vec{F} = m \vec{a}$) para la ecuación de movimiento de un oscilador armónico simple es:

$$m \frac{d^2 x}{dt^2} = -kx, \text{ o, pasando todo al mismo miembro: } m \frac{d^2 x}{dt^2} + kx = 0.$$

donde x representa el desplazamiento respecto a la posición de equilibrio, m la masa inercial y $-kx$ la fuerza restauradora. Cuando además actúa una fuerza amortiguadora (viscosa) proporcional a la velocidad $F_v = -cv = -c \frac{dx}{dt}$, la ecuación de movimiento queda:

$$m \frac{d^2 x}{dt^2} + c \frac{dx}{dt} + kx = 0$$

la cual, dividiendo por m , suele escribirse como:

$$\frac{d^2 x}{dt^2} + 2\beta \frac{dx}{dt} + \omega_0^2 x = 0$$

donde $2\beta=c/m$ es el factor de amortiguamiento y $\omega_0 = \sqrt{k/m}$ es la frecuencia de vibración libre. Demos un nombre a esta ecuación:

```
In[1]:= Clear[x, beta, omega0]
ecamort = x''[t] + 2 beta x'[t] + omega0^2 x[t]
```

```
Out[2]= omega0^2 x[t] + 2 beta x'[t] + x''[t]
```

La solución general de tal ecuación puede obtenerse como:

```
In[3]:= DSolve[ecamort == 0, x[t], t]
```

```
Out[3]= {{x[t] -> e^{t(-beta-sqrt(beta^2-omega0^2))} C[1] + e^{t(-beta+sqrt(beta^2-omega0^2))} C[2]}}
```


Dicha solución tiene comportamientos diferentes dependiendo de los valores de β y de ω_0 . Estudiemoslos caso por caso:

Oscilador subamortiguado $\beta < \omega_0$

Por ejemplo, tomemos $\beta=0.1$, $\omega_0 = 0.5$ y, como condición inicial: $x(0)=1$, $x'(0)=0$. La solución $x(t)$ y su gráfica vienen dadas por:

```
In[4]:= Clear[x]
subamort = DSolve[
  {ecamort == 0 /. {β -> 0.1, ω0 -> 0.5}, x[0] == 1, x'[0] == 0}, x, t]
Plot[x[t] /. subamort, {t, 0, 2 (2 Pi / 0.5)},
  AxesLabel -> {"t", "x[t]"}]
```


```
Out[5]= {{x -> (e-0.1 #1 (Cos[0.489898 #1] + 0.204124 Sin[0.489898 #1])) &}}
```



```
Out[6]= - Graphics -
```

Vemos que se trata de un movimiento vibratorio donde la amplitud disminuye paulatinamente con el tiempo hasta hacerse cero (en el límite $t \rightarrow \infty$). Veamos una representación gráfica del movimiento en el plano de fases:

```
In[7]:= ParametricPlot[Evaluate[{x[t], x'[t]} /. subamort],
  {t, 0, 2 (2 Pi / 0.5)}, AxesLabel -> {"x", "v[x]"}]
```



```
Out[7]= - Graphics -
```


Las órbitas son abiertas, lo cual indica que el movimiento no es periódico. También vemos que la trayectoria tiende al punto $(x(\infty)=0, v(\infty)=0)$.

Amortiguamiento crítico $\beta = \omega_0$

Por ejemplo, tomemos $\beta = \omega_0 = 0.1$ y, como condición inicial: $x(0)=1, x'(0)=0$. La solución $x(t)$ y su gráfica vienen dadas por:

```
In[8]:= Clear[x]
amortcrit = DSolve[
  {ecamort == 0 /. {β -> 0.1, ω0 -> 0.1}, x[0] == 1, x'[0] == 0}, x, t]
Plot[x[t] /. amortcrit, {t, 0, (2 Pi / 0.1)},
  AxesLabel -> {"t", "x[t]"}]
```


```
Out[9]= {{x -> (e-0.1 #1 (1. + 0.1 #1) &)}}
```



```
Out[10]= - Graphics -
```

Aquí el sistema no realiza ninguna oscilación y la posición va a cero paulatinamente con el tiempo ($x(\infty)=0$).
En el plano de fases:

```
In[11]:= ParametricPlot[Evaluate[{x[t], x'[t]} /. amortcrit],
  {t, 0, (2 Pi / 0.1)}, AxesLabel -> {"x", "v[x]"}]
```



```
Out[11]= - Graphics -
```


Vemos igualmente que la trayectoria tiende al punto ($x(\infty)=0$, $v(\infty)=0$) partiendo de ($x(0)=1$, $v(0)=0$)

Oscilador sobreamortiguado $\beta > \omega_0$

Por ejemplo, tomemos $\beta=0.3$, $\omega_0 = 0.1$ y, como condición inicial: $x(0)=1$, $x'(0)=0$. La solución $x(t)$, y su gráfica superpuesta a la de amortiguamiento crítico vienen dadas por:

```
In[12]:= Clear[x]
sobreamort = DSolve[
  {ecamort == 0 /. {β -> 0.3, ω0 -> 0.1}, x[0] == 1, x'[0] == 0}, x, t]
Plot[{x[t] /. sobreamort, x[t] /. amortcrit}, {t, 0, (2 Pi / 0.1)},
  AxesLabel -> {"t", "x[t]"}, AxesOrigin -> {0, 0}]
```


```
Out[13]:= {{x -> (-0.0303301 e-0.582843 #1 + 1.03033 e-0.0171573 #1 & )}}
```



```
Out[14]:= - Graphics -
```

La presencia de una mayor viscosidad hace que la posición vaya hacia cero más lentamente que en el caso crítico. La representación en el plano de fases es similar al caso crítico:

```
In[15]:= ParametricPlot[Evaluate[{x[t], x'[t]} /. sobreamort],
  {t, 0, (2 Pi / 0.1)}, AxesLabel -> {"x", "v[x]"}]
```



```
Out[15]:= - Graphics -
```

Ejercicio

Repita los casos anteriores para las condiciones iniciales $x(0)=0$, $x'(0)=1$ e interprete físicamente los resultados y las gráficas obtenidas.

■ Oscilador armónico forzado: pulsaciones y resonancia

Solución general

Si añadimos una nueva fuerza externa dependiente del tiempo $F(t)$ en la ecuación de movimiento de un oscilador armónico amortiguado obtenemos:

$$m \frac{d^2 x}{dt^2} + c \frac{dx}{dt} + kx = F(t)$$

la cual, dividiendo por m , suele escribirse como :

$$\frac{d^2 x}{dt^2} + 2\beta \frac{dx}{dt} + \omega_0^2 x = f(t)$$

Las fuerzas externas que suelen discutirse en este tipo de problemas son de tipo periódico $f(t+T)=f(t)$ como, por ejemplo:

$$f(t) = A \sin(\omega t),$$

donde A representa la amplitud de la acción externa $f(t)$ y ω representa su frecuencia. Un ejemplo de acción externa sobre un oscilador que todos hemos experimentado de pequeños puede ser "el empuje continuado y periódico sobre un columpio". Tomemos como ejemplo la siguiente ecuación:

```
In[16]:= Clear[x, β, ω0, ω, a]
          ecamforz = x''[t] + 2 β x'[t] + ω0^2 x[t] + a Sin[ω t]

Out[17]:= a Sin[t ω] + ω0^2 x[t] + 2 β x'[t] + x''[t]
```

La solución general de tal ecuación puede obtenerse como:

```
In[18]:= DSolve[ecamforz == 0, x[t], t]

Out[18]= { { x[t] -> e^{t (-β - √{β^2 - ω0^2})} C[1] + e^{t (-β + √{β^2 - ω0^2})} C[2] - \frac{1}{2 \sqrt{\beta^2 - \omega_0^2}} \left( a e^{t (-β + \sqrt{\beta^2 - \omega_0^2})} \left( -\frac{e^{-t (-β + \sqrt{\beta^2 - \omega_0^2})} \omega \cos[t \omega]}{(-β - i \omega + \sqrt{\beta^2 - \omega_0^2}) (-β + i \omega + \sqrt{\beta^2 - \omega_0^2})} - \frac{e^{-t (-β + \sqrt{\beta^2 - \omega_0^2})} (-β + \sqrt{\beta^2 - \omega_0^2}) \sin[t \omega]}{(-β - i \omega + \sqrt{\beta^2 - \omega_0^2}) (-β + i \omega + \sqrt{\beta^2 - \omega_0^2})} \right) \right) + \frac{1}{2 \sqrt{\beta^2 - \omega_0^2}} \left( a e^{t (-β - \sqrt{\beta^2 - \omega_0^2})} \left( -\frac{e^{t (β + \sqrt{\beta^2 - \omega_0^2})} \omega \cos[t \omega]}{(\beta - i \omega + \sqrt{\beta^2 - \omega_0^2}) (\beta + i \omega + \sqrt{\beta^2 - \omega_0^2})} + \frac{e^{t (β + \sqrt{\beta^2 - \omega_0^2})} (\beta + \sqrt{\beta^2 - \omega_0^2}) \sin[t \omega]}{(\beta - i \omega + \sqrt{\beta^2 - \omega_0^2}) (\beta + i \omega + \sqrt{\beta^2 - \omega_0^2})} \right) \right) } }
```

Dicha solución tiene comportamientos diferentes dependiendo de los valores de β , ω_0 y ω . Estudiemos algunos casos:

Oscilaciones forzadas no amortiguadas ($\beta=0$): pulsaciones y resonancia pura


Tomemos como caso particular $\omega_0 = 0.5$, $\omega = 2$, $a = 1$ y como condición inicial $x(0) = 1$, $x'(0) = 0$ y hagamos una representación gráfica de la trayectoria y de la órbita en el plano de fases

```


In[19]:= Clear[x]
forz = DSolve[{ecamforz == 0 /. {β -> 0, ω0 -> 0.5, ω -> 2, a -> 1},
  x[0] == 1, x'[0] == 0}, x, t]
Plot[x[t] /. forz, {t, 0, 20 Pi}, AxesLabel -> {"t", "x[t]"}]
ParametricPlot[Evaluate[{x[t], x'[t]} /. forz], {t, 0, 4 Pi}]

```

```
Out[20]= {{x -> (0.0666667 (15. Cos[0.5 #1] - 16. Sin[0.5 #1] + 4. Sin[2. #1]) &)}}
```



```
Out[21]= - Graphics -
```




```
Out[22]= - Graphics -
```

Vemos que la órbita es cerrada (indicativo de comportamiento periódico). No obstante, nótese la diferencia respecto al caso libre (no forzado), donde las órbitas son simplemente elipses. Aquí las órbitas presentan un comportamiento más rico, dependiendo de los valores de ω y ω_0 .


Para valores $\omega \approx \omega_0$ de la frecuencia externa ω muy cercanos a la frecuencia natural del sistema ω_0 , se da un comportamiento interesante en el sistema conocido como el fenómeno de los *latidos o pulsaciones*. Por ejemplo, tomemos $\omega_0 = 0.5$, $\omega = 0.55$:

```
In[23]:= forzpuls = DSolve[{ecamforz == 0 /. {β -> 0, ω0 -> 0.5, ω -> 0.55, a -> 1},
  x[0] == 1, x'[0] == 0}, x, t]
Plot[x[t] /. forzpuls, {t, 0, 200 Pi}, AxesLabel -> {"t", "x[t]"}]
ParametricPlot[Evaluate[{x[t], x'[t]} /. forzpuls], {t, 0, 50 Pi}]
```

```
Out[23]= {{x -> (0.047619
  (21. Cos[0.5 #1] - 440. Sin[0.5 #1] + 400. Sin[0.55 #1]) &)}}
```



```
Out[24]= - Graphics -
```




```
Out[25]= - Graphics -
```

Vemos como una oscilación de baja frecuencia (o periodo largo) $\omega_- = \omega - \omega_0 = 0.05$ "envuelve o modula" a una oscilación de alta frecuencia (o periodo corto) $\omega_+ = \omega + \omega_0 = 1.05$ (en comparación con la frecuencias originales). ¡Éste es el fundamento básico de la *frecuencia modulada* (FM), en la banda de frecuencias donde se suelen emitir las emisoras de radio musicales!

El comportamiento del sistema cambia radicalmente (deja de ser periódico) para la frecuencia externa "crítica" $\omega = \omega_0$, es decir, cuando la frecuencia de la acción externa ω coincide con la frecuencia natural del sistema masa-resorte ω_0 . En efecto, tomemos $\omega = \omega_0 = 0.5$

```
In[26]:= Clear[x]
forzcrit = DSolve[{ecamforz == 0 /. {β -> 0, ω0 -> 0.5, ω -> 0.5, a -> 1},
  x[0] == 1, x'[0] == 0}, x, t]
Plot[x[t] /. forzcrit, {t, 0, 5 Pi}, AxesLabel -> {"t", "x[t]"}]
```


```
Out[27]= {{x -> (Cos[0.5 #1] - 2. Sin[0.5 #1] + Cos[0.5 #1] #1 &)}}
```



```
Out[28]= - Graphics -
```

Vemos que la amplitud de las oscilaciones crece indefinidamente con el tiempo. En el plano de fases tenemos:

```
In[29]:= ParametricPlot[Evaluate[{x[t], x'[t]} /. forzcrit], {t, 0, 20 Pi}]
```



```
Out[29]= - Graphics -
```

una órbita en espiral (hacia afuera). Este fenómeno se denomina: *resonancia pura*. Lo que suele ocurrir en los sistemas físicos reales en condiciones de resonancia pura es que el crecimiento indefinido de la amplitud de las oscilaciones provoca una "rotura" del sistema al sobrepasar el límite elástico. En obras de ingeniería civil, como los puentes, es conveniente asegurarse de que la frecuencia natural de oscilación del puente sea suficientemente distinta de la frecuencia de las acciones externas (ritmo de vehículos, peatones, viento local, etc), lo cual se logra usando unos materiales y una geometría adecuada.

Oscilaciones forzadas amortiguadas ($\beta \neq 0$): resonancia

Tomemos como caso particular $\beta = 0.1$, $\omega_0 = 0.5$, $\omega = 2$, $a = 1$ y como condición inicial $x(0) = 1$, $x'(0) = 0$ y hagamos una representación gráfica de la trayectoria y de la órbita en el plano de fases


```


In[30]= Clear[x]
forzamort =
DSolve[{ecamforz == 0 /. {β -> 0.1, ω0 -> 0.5, ω -> 2, a -> 1},
x[0] == 1, x'[0] == 0}, x, t]
Plot[x[t] /. forzamort, {t, 0, 20 Pi}, AxesLabel -> {"t", "x[t]"}]
ParametricPlot[Evaluate[{x[t], x'[t]} /. forzamort],
{t, 0, 40 Pi}, AxesLabel -> {"x", "v[x]"}]

```

```

Out[31]= {{x -> ((0.0000292963 + 5.99706 × 10-23 i) e-0.1 #1
(33174. Cos[0.489898 #1] + 960. e0.1 #1 Cos[2. #1] -
29970.7 Sin[0.489898 #1] + 9000. e0.1 #1 Sin[2. #1]) &)}}


```



```

Out[32]= - Graphics -

```



```


Out[33]= - Graphics -

```


Vemos que la órbita no es cerrada pero que se aproxima asintóticamente a una órbita elíptica. En el gráfico observamos un "régimen o estado transitorio", que dura unos 20 o 30 segundos, tras el cual la oscilación se estabiliza, llegando a un "régimen o estado estacionario", que se corresponde con la solución particular de la ecuación no homogénea. Para ω y ω_0 fijas, la duración del transitorio depende de la amortiguación β . Por ejemplo, si tomamos una amortiguación mayor como, por ejemplo $\beta=0.3$, obtenemos que:

```
In[34]= Clear[x]
forzamort =
DSolve[{ecamforz == 0 /. {β -> 0.3, ω0 -> 0.5, ω -> 2, a -> 1},
x[0] == 1, x'[0] == 0}, x, t]
forzamortplot = Plot[x[t] /. forzamort, {t, 0, 20 Pi},
AxesLabel -> {"t", "x[t]"}]
forzamortfase = ParametricPlot[Evaluate[{x[t], x'[t]} /. forzamort],
{t, 0, 40 Pi}, AxesLabel -> {"x", "v[x]"}]
```

```
Out[35]= {{x -> (0.000483793 e-0.3 #1 (1907. Cos[0.4 #1] + 160. e0.3 #1 Cos[2. #1] -
1069.75 Sin[0.4 #1] + 500. e0.3 #1 Sin[2. #1]) & )}}
```



```
Out[36]= - Graphics -
```



```
Out[37]= - Graphics -
```

se alcanza antes el régimen estacionario (a los 10 segundos aproximadamente...). Esto está de acuerdo con nuestra experiencia.

Para $\omega = \omega_0$ tenemos una *resonancia* (no "pura"), caracterizada porque la amplitud de las oscilaciones alcanza un máximo, sin crecer indefinidamente con el tiempo (como sucede en la resonancia pura), debido a la presencia de amortiguamiento. Tomemos el mismo amortiguamiento que en el caso anterior $\beta=0.3$, y $\omega_0 = 0.5 = \omega$. Para estos valores tenemos:

```


In[38]:= Clear[x]
forzamortcrit =
  DSolve[{ecamforz == 0 /. {β -> 0.3, ω0 -> 0.5, ω -> 0.5, a -> 1},
    x[0] == 1, x'[0] == 0}, x, t]
forzamortplotcrit = Plot[x[t] /. forzamortcrit,
  {t, 0, 20 Pi}, AxesLabel -> {"t", "x[t]"}]
forzamortfasecrit = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. forzamortcrit],
  {t, 0, 40 Pi}, AxesLabel -> {"x", "v[x]"}]

```


```

Out[39]= {{x -> (0.333333 e-0.3 #1
  (-7. Cos[0.4 #1] + 10. e0.3 #1 Cos[0.5 #1] - 5.25 Sin[0.4 #1]) &)} }

```



```
Out[40]= - Graphics -
```




```
Out[41]= - Graphics -
```

Superpongamos las gráficas de la solución no crítica $\{\beta > 0.3, \omega_0 > 0.5, \omega > 2, a > 1\}$ con la crítica $\{\beta > 0.3, \omega_0 > 0.5, \omega > 0.5, a > 1\}$


```

In[42]:= Show[{forzamortplot, forzamortplotcrit}, PlotRange -> {-3.5, 3.5}]
Show[forzamortfase, forzamortfasecrit]

```



```
Out[42]= - Graphics -
```



```
Out[43]= - Graphics -
```

Vemos que la amplitud de las oscilaciones es, con diferencia, mayor en el caso crítico. También lo es la velocidad, tal y como muestra el diagrama de fases. Como se ha comentado anteriormente, el fenómeno de la resonancia tiene una doble vertiente. En su vertiente "positiva", este fenómeno es explotado, por ejemplo, para SINTONIZAR ciertas frecuencias próximas a la frecuencia natural propia de ciertos cristales o ciertos circuitos (tal y como sucede en nuestro aparato de radio). En su vertiente "negativa", el fenómeno de la resonancia puede provocar la ROTURA de ciertas estructuras sometidas a una acción externa periódica de frecuencia próxima a la frecuencia natural de la estructura, debido a las fuertes oscilaciones provocadas en estas condiciones críticas.

Ejercicio

Repita el caso anterior para un amortiguamiento fuerte $\beta=1$. ¿Cómo afecta el amortiguamiento a la resonancia?, es decir, ¿se aprecia mejor la resonancia para el oscilador sobreamortiguado que para el subamortiguado o al revés?.


Fuerzas externas definidas a trozos

Estudiemos el oscilador armónico amortiguado y forzado:

$$\frac{d^2 x}{dt^2} + 2\beta \frac{dx}{dt} + \omega_0^2 x = f(t)$$

para funciones más generales que las de tipo sinusoidal. Por ejemplo, centrémonos en funciones definidas a trozos como, por ejemplo, la función salto

```
In[44]:= salto[t_] := If[t >= 1, 1000, 0]
Plot[salto[t], {t, 0, 2}]
```



Out[45]= - Graphics -

que representa una fuerza que no actúa hasta el instante $t=1$, en el cual adquiere un valor constante $=1000$. Este tipo de fuerzas aparecen, por ejemplo, al accionar repentinamente un "interruptor"... Veamos cuál es el efecto de dicha fuerza sobre un oscilador


```
In[46]:= Clear[x, beta, omega0]
ecamsalto = x''[t] + 2 beta x'[t] + omega0^2 x[t] + salto[t]
```

```
Out[47]= If[t >= 1, 1000, 0] + omega0^2 x[t] + 2 beta x'[t] + x''[t]
```


Para este tipo de funciones, *Mathematica* necesita de recursos numéricos para su resolución. Por ejemplo, tomemos amortiguamiento $\beta=0$, frecuencia natural $\omega_0 = 30$ y condición inicial $x[0]=1, x'[0]=0$ y resolvamos numéricamente en el intervalo $0 \leq t \leq 2$

```
In[48]= Clear[x]
chupinazo = NDSolve[{ecamsalto == 0 /. {β -> 0, ω0 -> 30},
  x[0] == 1, x'[0] == 0}, x, {t, 0, 2}]
Plot[x[t] /. chupinazo, {t, 0, 2}, AxesLabel -> {"t", "x[t]"}]
ParametricPlot[Evaluate[{x[t], x'[t]} /. chupinazo], {t, 0, 2}]
```

```
Out[49]= {{x -> InterpolatingFunction[{{0., 2.}}, <>]}}
```



```
Out[50]= - Graphics -
```



```
Out[51]= - Graphics -
```

Vemos como, a partir del instante $t=1$, el comportamiento de la solución cambia de forma continua pero repentina. Por ejemplo, la órbita en el espacio de las fases (elipses) se desplaza hacia la izquierda y aumenta de tamaño; esto quiere decir que, tras la acción de la fuerza en $t=1$, el oscilador pasa más tiempo en los valores $x < 0$ que en $x > 0$, además de aumentar su velocidad.

Ejercicio

Estudiar el comportamiento del oscilador anterior en el intervalo $0 \leq t \leq 3$ para un pulso

$$p(t) = \begin{cases} 10000, & \text{si } 1 \leq t \leq 2 \\ 0, & \text{en otro caso} \end{cases}$$

Interprete los resultados.

Ayuda: recuerde el comando **Which**[condicion1,proceso1,condición2,proceso2,...] para definir funciones a trozos.

Ejercicio

Estudie la solución del oscilador armónico simple de frecuencia 1 sometido a una función periódica del tipo "diente de sierra":

$$f[t]=t-\text{Round}[t]$$

en el intervalo $0 \leq t \leq 20$

■ Oscilador anarmónico

Discusión general

Estudiaremos en esta sección el caso más complicado (no lineal) de una fuerza recuperadora no lineal del tipo $F(x)=-kx-\delta x^3$. Para ganar intuición sobre el tipo de comportamiento que esperamos, hagamos una representación gráfica de dicha fuerza para tres casos distintos y valores particulares de los parámetros:


1) Armónico ($\delta=0$): $F_0(x) = -x$

2) Anarmonicidad débil ($\delta \ll k$): $F_1(x) = -x - \frac{1}{10}x^3$

3) Anarmonicidad fuerte ($\delta \approx k$): $F_2(x) = -x - x^3$

4) Anarmonicidad atractivo-repulsiva ($k < 0, \delta > 0$): $F_3(x) = x - x^3$

```
In[52]:= << Graphics`Legend`
Plot[{-x, -x - 0.1 x^3, -x - x^3, x - x^3}, {x, -2, 2}, PlotStyle ->
  {GrayLevel[0], Dashing[ {.01}], Dashing[ {.03}], Dashing[ {.05}]},
PlotLegend -> {"F0(x)", "F1(x)", "F2(x)", "F3(x)"},
AxesLabel -> {"x", "F(x)"}]
```



Out[53]= - Graphics -

Mientras que en los tres primeros casos el único punto de equilibrio ($F=0$) es $x=0$, para el cuarto caso tenemos tres puntos de equilibrio: $x=0,1,-1$.

Estudiemos las soluciones de la ecuación de un oscilador anarmónico amortiguado sometido a una fuerza externa dependiente del tiempo $F_e(t)$. Dicha ecuación admite una expresión general del tipo:

$$m \frac{d^2 x}{dt^2} + c \frac{dx}{dt} + kx + \delta x^3 = F_e(t)$$

la cual, dividiendo por m , suele escribirse como :

$$\frac{d^2 x}{dt^2} + 2\beta \frac{dx}{dt} + \omega_0^2 x + \gamma x^3 = f_e(t)$$

Tomemos por ejemplo $f_e(t)=a \operatorname{sen}(\omega t)$ y demos un nombre a esta ecuación


```
In[54]:= Clear[x, beta, omega0, omega, gamma, a]
ecanamfor = x''[t] + 2 beta x'[t] + omega0^2 x[t] + gamma x[t]^3 + a Sin[omega t]
Out[55]= a Sin[t omega] + omega0^2 x[t] + gamma x[t]^3 + 2 beta x'[t] + x''[t]
```

Vemos que se trata de una ecuación no lineal de segundo orden. Necesitaremos técnicas numéricas para su resolución. Tenemos una casuística bastante rica dependiendo de los parámetros $\beta, \omega_0, \omega, \delta, a$. Tratemos algunos casos:


Oscilador anarmónico libre ($\beta=0=a$)

Comencemos por perturbar débilmente ($\gamma \ll \omega_0^2$) el oscilador armónico. Representemos primeramente la solución del caso armónico ($\gamma=0$) para $\omega_0 = 1$ y condiciones iniciales $x(0)=1, x'(0)=0$:

```
In[56]:= armonico = DSolve[{x''[t] == -x[t], x[0] == 1, x'[0] == 0}, x, t]
xtar = Plot[x[t] /. armonico, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[0, 1, 0]}]
fasear = ParametricPlot[Evaluate[{x[t], x'[t]} /. armonico],
  {t, 0, 2 Pi}, AxesLabel -> {"x", "v[x]"},
  PlotStyle -> {RGBColor[0, 1, 0]}]
Out[56]= {{x -> (Cos[#1] &)}}
```



Out[57]= - Graphics -



Out[58]= - Graphics -


y perturbemos débilmente con $\gamma=0.1$

```


In[59]:= Clear[x]
anarmdebil =
NDSolve[{ecanamfor == 0 /. { $\beta \rightarrow 0$ ,  $\omega_0 \rightarrow 1$ ,  $\omega \rightarrow 0$ ,  $a \rightarrow 0$ ,  $\gamma \rightarrow 0.1$ },
  x[0] == 1, x'[0] == 0}, x, {t, 0, 20 Pi}]
xtanardeb = Plot[x[t] /. anarmdebil, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[1, 0, 0]}]
faseanardeb = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. anarmdebil[[1]]],
  {t, 0, 2 Pi}, PlotStyle -> {RGBColor[1, 0, 0]}]

```

```
Out[60]= {{x -> InterpolatingFunction[{{0., 62.8319}}, <>]}}
```




```
Out[61]= - Graphics -
```




```
Out[62]= - Graphics -
```

Superpongamos ambos gráficos:


```
In[63]:= Show[xtar, xtandardeb]  
Show[fasear, faseanardeb]
```



```
Out[63]= - Graphics -
```



```
Out[64]= - Graphics -
```

Vemos que el efecto de la perturbación anarmónica $\gamma=0.1$ es generar un pequeño adelanto de la oscilación anarmónica (roja) respecto de la armónica (verde) debido a un pequeño incremento de la velocidad en las oscilaciones anarmónicas. Veamos qué pasa para una anarmonicidad fuerte como $\gamma=1$

```


In[65]:= Clear[x]
anarmfuerte =
NDSolve[{ecanamfor == 0 /. {β -> 0, ω0 -> 1, ω -> 0, a -> 0, γ -> 1},
  x[0] == 1, x'[0] == 0}, x, {t, 0, 20 Pi}]
xtanarfu = Plot[x[t] /. anarmfuerte, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[1, 0, 0]},
  DisplayFunction -> Identity]
faseanarfu = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. anarmfuerte[[1]]], {t, 0, 2 Pi},
  PlotStyle -> {RGBColor[1, 0, 0]}, DisplayFunction -> Identity]
Show[xtar, xtanarfu]
Show[fasear, faseanarfu]

```


```
Out[66]= {{x -> InterpolatingFunction[{{0., 62.8319}}, <>]}}
```

```
Out[67]= - Graphics -
```

```
Out[68]= - Graphics -
```



```
Out[69]= - Graphics -
```



```
Out[70]= - Graphics -
```


Aquí el adelanto de la oscilación anarmónica respecto a la armónica es mucho mayor que antes, debido a que la diferencia de velocidades es mayor (¿por qué?), tal y como se observa en el diagrama de fases.

Ejercicio

Repita los cálculos para una anarmonicidad muy fuerte $\gamma=10$ y condiciones iniciales $x(0)=0, x'(0)=1$. Compare con el caso armónico $\gamma=0$.

Por último, veamos qué sucede para una fuerza atractivo-repulsiva con $\omega_0^2 = -1 = -\gamma$, es decir $F_3(x) \propto x - x^3$, con tres puntos de equilibrio: $x=0$ (inestable), $x=\pm 1$ (estable). El potencial $U(x) = -\int (x - x^3) dx = \frac{x^4}{4} - \frac{x^2}{2}$ tiene la forma:

```
In[71]= Plot[x^4/4 - x^2/2, {x, -2, 2}]
```



```
Out[71]= - Graphics -
```

de "sombrero mejicano" con dos "pozos" en $x=\pm 1$ y una "cima" en $x=0$. Tengamos en cuenta tres posibles condiciones iniciales:

- 1) $x(0)=-0.5$, $x'(0)=0$: la partícula parte del reposo en los alrededores del "pozo" izquierdo.
- 2) $x(0)=0.5$, $x'(0)=0$: la partícula parte del reposo en los alrededores del "pozo" derecho
- 3) $x(0)=0$, $x'(0)=0.1$: la partícula parte de la "cima" con una pequeña velocidad inicial hacia la derecha

```


In[72]:= Clear[x]
anarmrepatr1 =
NDSolve[{ecanamfor == 0 /. {β -> 0, ω0 -> I, ω -> 0, a -> 0, γ -> 1},
  x[0] == -0.5, x'[0] == 0}, x, {t, 0, 2 Pi}]
fase1 = ParametricPlot[Evaluate[{x[t], x'[t]} /. anarmrepatr1[[1]]],
  {t, 0, 2 Pi}, PlotStyle -> {{RGBColor[1, 0, 0]}},
  AxesOrigin -> {0, 0}]

Clear[x]
anarmrepatr2 =
NDSolve[{ecanamfor == 0 /. {β -> 0, ω0 -> I, ω -> 0, a -> 0, γ -> 1},
  x[0] == 0.5, x'[0] == 0}, x, {t, 0, 2 Pi}]
fase2 = ParametricPlot[Evaluate[{x[t], x'[t]} /. anarmrepatr2[[1]]],
  {t, 0, 2 Pi}, PlotStyle -> {RGBColor[0, 1, 0]}, AxesOrigin -> {0, 0}]

Clear[x]
anarmrepatr3 =
NDSolve[{ecanamfor == 0 /. {β -> 0, ω0 -> I, ω -> 0, a -> 0, γ -> 1},
  x[0] == 0, x'[0] == 0.1}, x, {t, 0, 6 Pi}]
fase3 = ParametricPlot[Evaluate[{x[t], x'[t]} /. anarmrepatr3[[1]]],
  {t, 0, 6 Pi}, PlotStyle -> {RGBColor[0, 0, 1]}, AxesOrigin -> {0, 0}]


Out[73]= {{x -> InterpolatingFunction[{{0., 6.28319}}, <>]}}

```




Out[74]= - Graphics -

```
Out[76]= {{x -> InterpolatingFunction[{{0., 6.28319}}, <>]}}
```



Out[77]= - Graphics -


```
Out[79]= {{x -> InterpolatingFunction[{{0., 18.8496}}, <>]}}
```



Out[80]= - Graphics -

Combinemos estas tres órbitas en un solo gráfico:

```
In[81]:= Show[{fase1, fase2, fase3}]
```



```
Out[81]= - Graphics -
```

Vemos que, en el caso 1), la partícula oscila alrededor del mínimo de potencial $x=-1$ (posición de equilibrio estable). Para el caso 2) la partícula oscila en los alrededores de $x=1$. En el tercer caso la partícula oscila entre pozo y pozo, pasando de uno a otro de forma periódica (observe como la velocidad disminuye en las cercanías de la "cima" $x=0$ ¿por qué?).

Ejercicio

Repita el ejercicio anterior para las siguientes condiciones iniciales:

- 1) $x(0)=-1, x'(0)=1$: la partícula parte de la posición de equilibrio $x=-1$ con velocidad $x'=1$.
- 2) $x(0)=0.5, x'(0)=0$: la partícula parte de la posición de equilibrio $x=1$ con velocidad $x'=-1$.
- 3) $x(0)=0, x'(0)=-0.2$: la partícula parte de la "cima" con una pequeña velocidad inicial hacia la izquierda

Ecuación de Duffing ($\omega_0 = \gamma = 1$)

Veamos cuál es el efecto de añadir un término anarmónico x^3 al oscilador armónico amortiguado forzado estudiado anteriormente. Tomemos $\beta=0.1, \omega_0 = 1, \omega = 2, a = 1$ y representemos primeramente la solución del caso armónico ($\gamma=0$) con condiciones iniciales $x(0)=1, x'(0)=0$:

```


In[82]:= Clear[x]
armoamorfor =
DSolve[{ecanamfor == 0 /. {β -> 0.1, ω0 -> 1, ω -> 2, a -> 1, γ -> 0},
  x[0] == 1, x'[0] == 0}, x, t]
xtaramf = Plot[x[t] /. armoamorfor, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[0, 1, 0]}]
fasearamf = ParametricPlot[Evaluate[{x[t], x'[t]} /. armoamorfor],
  {t, 0, 20 Pi}, AxesLabel -> {"x", "v[x]"},
  PlotStyle -> {RGBColor[0, 1, 0]}]

```

```

Out[83]= {{x -> ((0.0000661638 - 6.24521 × 10-22 i) e-0.1 #1
  (14454. Cos[0.994987 #1] + 660. e0.1 #1 Cos[2. #1] -
  8497.19 Sin[0.994987 #1] + 4950. e0.1 #1 Sin[2. #1]) &)}}


```



```

Out[84]= - Graphics -

```



```


Out[85]= - Graphics -

```


y perturbemos con $\gamma=1$

```
In[86]:= Clear[x]
anarmoamorfor =
NDSolve[{ecanamfor == 0 /. { $\beta \rightarrow 0.1$ ,  $\omega_0 \rightarrow 1$ ,  $\omega \rightarrow 2$ ,  $a \rightarrow 1$ ,  $\gamma \rightarrow 1$ },
  x[0] == 1, x'[0] == 0}, x, {t, 0, 20 Pi}]
xtanaramf = Plot[x[t] /. anarmoamorfor, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[1, 0, 0]}]
faseanaramf = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. anarmoamorfor], {t, 0, 20 Pi},
  AxesLabel -> {"x", "v[x]"}, PlotStyle -> {RGBColor[1, 0, 0]}]
```

```
Out[87]= {{x -> InterpolatingFunction[{{0., 62.8319}}, <>]}}
```




```
Out[88]= - Graphics -
```




```
Out[89]= - Graphics -
```

```
In[90]:= Show[xtaramf, xtanaramf]
Show[fasearamf, faseanaramf]
```



```
Out[90]= - Graphics -
```



```
Out[91]= - Graphics -
```

Vemos que el efecto de la perturbación anarmónica $\gamma=1$ no es importante en el estado estacionario (aunque si en el transitorio).

Ejercicio

Estudie la ecuación de Duffing para $\beta=0.2$, $\omega_0 = 1 = \gamma$, $\omega = 2$, $a = 1$ y condiciones iniciales $x(0)=0, x'(0)=1$. Compare con el caso armónico $\gamma=0$.

■ El péndulo

Discusión general

Consideremos la ecuación general de movimiento de un péndulo amortiguado y forzado:

$$\ddot{\phi} + 2\beta\dot{\phi} + \omega_0^2 \sin(\phi) = f(t).$$

Vemos que se trata de una ecuación no lineal de segundo orden. Tomemos por ejemplo $f(t)=a \sin(\omega t)$ e introduzcamos la ecuación:

```
In[92]:= Clear[x, beta, omega0, omega, gamma, a]
          ependamfor = x''[t] + 2 beta x'[t] + omega0^2 Sin[x[t]] + a Sin[omega t]
Out[93]:= a Sin[t omega] + omega0^2 Sin[x[t]] + 2 beta x'[t] + x''[t]
```

Consideremos diferentes casos:

Péndulo simple ($\beta=0=a$)


Soltemos el péndulo desde la horizontal, es decir, desde un ángulo inicial de 90 grados ($x[0]==\pi/2$) y en reposo ($x'[0]==0$)


```


In[94]:= Clear[x]
pendulosimple =
  NDSolve[{ecpendamfor == 0 /. { $\beta$  -> 0,  $\omega_0$  -> 1,  $\omega$  -> 0, a -> 0},
    x[0] == Pi/2, x'[0] == 0}, x, {t, 0, 10 Pi}]
xtpendsimp = Plot[x[t] /. pendulosimple, {t, 0, 10 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[1, 0, 0]}]
fasependsimp = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. pendulosimple[[1]],
    {t, 0, 3 Pi}, PlotStyle -> {RGBColor[1, 0, 0]}]

```

```
Out[95]= {{x -> InterpolatingFunction[{{0., 31.4159}}, <>]}}
```



```
Out[96]= - Graphics -
```




```
Out[97]= - Graphics -
```


Comparemos con un oscilador armónico simple con idénticas condiciones iniciales

```
In[98]:= Clear[armonico, xtar, fasear]
armonico = DSolve[{x''[t] == -x[t], x[0] == Pi/2, x'[0] == 0}, x, t]
xtar = Plot[x[t] /. armonico, {t, 0, 10 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[0, 1, 0]}]
fasear = ParametricPlot[Evaluate[{x[t], x'[t]} /. armonico],
  {t, 0, 3 Pi}, AxesLabel -> {"x", "v[x]"},
  PlotStyle -> {RGBColor[0, 1, 0]}]
```

```
Out[99]= {{x -> (1/2) Pi Cos[#1] &}}
```




```
Out[100]= - Graphics -
```




```
Out[101]= - Graphics -
```

Superpongamos las gráficas del péndulo y del oscilador armónico simples:

```
In[102]:= Show[xtpendsimp, xtar]
Show[fasependsimp, fasear]
```



```
Out[102]= - Graphics -
```



```
Out[103]= - Graphics -
```

Vemos que, a idénticas condiciones iniciales, el periodo del péndulo simple es mayor que el del oscilador armónico simple. Sin embargo, su velocidad es menor, como muestra el diagrama de fases. Estas diferencias entre el oscilador armónico y el péndulo se hacen cada vez más pequeñas conforme la amplitud de las oscilaciones del péndulo se aproxima más a cero, ¿por qué?

Ejercicio

Repetid los cálculos anteriores para las condiciones iniciales:

$$x[0]=\pi/4, x'[0]=0$$

¿Es cierto que las diferencias entre las soluciones del péndulo simple y del oscilador armónico simple son menores ahora (menor amplitud) que para el caso anterior $x[0]=\pi/2, x'[0]=0$ (mayor amplitud)?

Ejercicio

Para un oscilador armónico simple, se demuestra que el periodo T no depende de la amplitud A de las oscilaciones. Compruebe que esto es cierto superponiendo las gráficas de las soluciones del oscilador armónico simple $x''(t) + \omega_0^2 x(t) = 0$ con condiciones iniciales a) $x[0]=\pi/2, x'[0]=0$ y b) $x[0]=\pi/4, x'[0]=0$, tomando $\omega_0^2 = 1$.

Repita el ejercicio para el péndulo simple. ¿Depende ahora el periodo de la amplitud?. ¿Aumenta el periodo con la amplitud o disminuye?

Péndulo amortiguado


Tomemos como factor de amortiguamiento $\beta=0.1$ y como condición inicial: "péndulo horizontal en reposo"

```


In[104]:= Clear[x]
penduloamort =
  NDSolve[{ecpendamfor == 0 /. { $\beta$  -> 0.1,  $\omega_0$  -> 1,  $\omega$  -> 0, a -> 0},
    x[0] == Pi/2, x'[0] == 0}, x, {t, 0, 10 Pi}]
xtpendmort = Plot[x[t] /. penduloamort, {t, 0, 10 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[1, 0, 0]}]
fasependmort = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. penduloamort[[1]]],
  {t, 0, 10 Pi}, PlotStyle -> {RGBColor[1, 0, 0]}]

```

```
Out[105]= {{x -> InterpolatingFunction[{{0., 31.4159}}, <>]}}
```



```
Out[106]= - Graphics -
```



```
Out[107]= - Graphics -
```

Vemos que la amplitud de las oscilaciones disminuye paulatinamente en el tiempo, al igual que la velocidad, tendiendo ambos asintóticamente a cero (posición de equilibrio)

Ejercicio

Compare el péndulo amortiguado anterior con el oscilador armónico amortiguado. ¿En qué caso es el amortiguamiento más rápido?

Péndulo amortiguado forzado


Tomemos como factor de amortiguamiento $\beta=0.1$, como frecuencia natural $\omega_0^2 = 1$, como frecuencia externa $\omega=2$ y amplitud $a=1$, y como condición inicial: "péndulo a 45 grados de la vertical y en reposo"

```


In[108]:= Clear[x]
penduloamfor =
  NDSolve[{ecpendamfor == 0 /. {β -> 0.1, ω0 -> 1, ω -> 2, a -> 1},
    x[0] == Pi/4, x'[0] == 0}, x, {t, 0, 20 Pi}]
xtpendamfor = Plot[x[t] /. penduloamfor, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[1, 0, 0]}]
fasependamfor = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. penduloamfor[[1]]],
  {t, 0, 20 Pi}, PlotStyle -> {RGBColor[1, 0, 0]}]

```

```
Out[109]= {{x -> InterpolatingFunction[{{0., 62.8319}}, <>]}}
```



```
Out[110]= - Graphics -
```



```
Out[111]= - Graphics -
```


Al igual que para el oscilador armónico forzado, observamos la presencia de un régimen estacionario tras un periodo transitorio. Veamos cómo se comporta el péndulo para la resonancia $\omega_0^2 = \omega^2 = 1$

```


In[112]= Clear[x]
penduloamforcrit =
  NDSolve[{ecpendamfor == 0 /. { $\beta$  -> 0.1,  $\omega_0$  -> 1,  $\omega$  -> 1, a -> 1},
    x[0] == Pi/4, x'[0] == 0}, x, {t, 0, 20 Pi}]
xtpendamforcrit = Plot[x[t] /. penduloamforcrit, {t, 0, 20 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[0, 1, 0]}]
fasependamforcrit = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. penduloamforcrit[[1]]],
  {t, 0, 20 Pi}, PlotStyle -> {RGBColor[0, 1, 0]}]

```

```
Out[113]= {{x -> InterpolatingFunction[{{0., 62.8319}}, <>]}}
```



```
Out[114]= - Graphics -
```



```
Out[115]= - Graphics -
```

Observamos cómo la amplitud de las oscilaciones es mayor en el caso crítico $\omega_0 = \omega = 1$ (con una amplitud ≈ 2) que en caso anterior $\omega_0 = 1, \omega = 2$ (con amplitud ≈ 0.25). Este es el fenómeno de la RESONANCIA.


Veamos que sucede si eliminamos el amortiguamiento:

```


In[116]:= Clear[x]
penduloforcrit =
  NDSolve[{ecpendamfor == 0 /. { $\beta$  -> 0,  $\omega_0$  -> 1,  $\omega$  -> 1, a -> 1},
    x[0] == Pi/4, x'[0] == 0}, x, {t, 0, 40 Pi}, MaxSteps -> 3000]
xtpendforcrit = Plot[x[t] /. penduloforcrit, {t, 0, 40 Pi},
  AxesLabel -> {"t", "x[t]"}, PlotStyle -> {RGBColor[0, 0, 1]}]
fasependforcrit = ParametricPlot[
  Evaluate[{x[t], x'[t]} /. penduloforcrit[[1]]],
  {t, 0, 40 Pi}, PlotStyle -> {RGBColor[0, 0, 1]}]

```

```
Out[117]= {{x -> InterpolatingFunction[{{0., 125.664}}, <>]}}
```



```
Out[118]= - Graphics -
```



```
Out[119]= - Graphics -
```

La solución presenta un aspecto "caótico"...

Ejercicio

Represente la trayectoria y el diagrama de fases de un péndulo con frecuencia natural $\omega_0=1$, sometido a una fuerza externa del tipo $f(t)=0.5 \operatorname{sen}(3t)$ y a un amortiguamiento de $\beta=0.2$, que parte del reposo en la posición vertical ($x[0]=0$).